

**Remarques :**

- Aucun document ni machine électronique n'est permis à l'exception de la carte de référence de Scheme.
- Les questions peuvent être résolues de façon indépendante. Il est possible, voire même utile, pour répondre à une question, d'utiliser les fonctions qui sont l'objet des questions précédentes.
- La clarté des réponses et la présentation des programmes seront appréciées. Toutes les fonctions apparaissant dans les réponses doivent être accompagnées de leur spécification (lorsqu'elle n'est pas donnée dans le sujet).
- La note tiendra compte de l'efficacité de vos programmes.
- Le barème (total sur 50) n'est donné qu'à titre indicatif.

## Exercice I : fonctions sur les listes

**Question 1** (4 points) :

Écrire une définition et une expression de test de la fonction `tous-egaux-a?` de spécification :

*;; tous-egaux-a? : LISTE[alpha] \* alpha -> bool  
;; (tous-egaux-a? L elt) rend #t ssi tous les éléments de la liste «L» sont égaux  
;; à la valeur de «elt». La fonction «tous-egaux-a?» rend, bien sûr, #t lorsque la  
;; liste «L» est vide.*

```
(define (tous-egaux-a? L elt)
  (if (pair? L)
      (and (equal? (car L) elt)
            (tous-egaux-a? (cdr L) elt) )
      #t))

(verifier tous-egaux-a?
 (tous-egaux-a? '(a a a a) 'a) == #t
 (tous-egaux-a? '(a b a a) 'a) == #f
 (tous-egaux-a? '() 'a) == #t )
```

**Question 2** (4 points) :

En utilisant la fonction `tous-egaux-a?`, écrire une définition de la fonction `tous-egaux?` spécifiée par :

```
;; ; tous-egaux? : LISTE[alpha] -> bool
;; ; (tous-egaux? L) rend #t ssi tous les éléments de la liste «L» sont égaux.
;; ; La fonction «tous-egaux?» rend, bien sûr, #t lorsque la liste «L» est vide.
```

Quelle est la valeur de l'application `(tous-egaux? '(a))` ?

```
(define (tous-egaux? L)
  (or (not (pair? L)) (tous-egaux-a? (cdr L) (car L)) ) )

(tous-egaux? '(a)) → #T
```

**Question 3** (5 points) :

Écrire une définition de l'itérateur `verifient-tous?` spécifié par :

```
;; ; verifient-tous? : (alpha -> bool) * LISTE[alpha] -> bool
;; ; (verifient-tous? test? L) rend #t ssi tous les éléments de la liste «L»
;; ; satisfont «test?». La fonction «verifient-tous?» rend bien sûr #t lorsque
;; ; la liste donnée est vide.
```

Par exemple :

```
(verifient-tous? odd? '(1 3 5)) → #T
(verifient-tous? odd? '(1 2 5)) → #F
(verifient-tous? odd? '()) → #T
```

```
(define (verifient-tous? test? L)
  (if (pair? L)
      (and (test? (car L))
           (verifient-tous? test? (cdr L)) )
      #t))
```

**Question 4** (4 points) :

En utilisant l'itérateur `verifient-tous?` précédent, écrire une spécification, une définition et une expression de test du prédicat `toutes-non-vides?` qui a pour donnée une liste de listes et qui vérifie que tous les éléments de la liste sont des listes non vides.

```

; ; ; toutes-non-vides? : LISTE[LISTE[alpha]] -> bool
; ; ; (toutes-non-vides? LL) rend #t ssi tous les éléments de la liste «LL» sont
; ; ; des listes non vides
(define (toutes-non-vides? LL)
  (verifient-tous? pair? LL))

(verifier toutes-non-vides?
  (toutes-non-vides? '((b) (b c))) == #t
  (toutes-non-vides? '((b) (b c) ())) == #f )

```

**Question 5** (5 points) :

En utilisant l'itérateur `verifient-tous?` précédent, écrire une autre définition de la fonction `tous-egaux-a?`.

```

(define (tous-egaux-a? L elt)
  ; ; equal-elt? : alpha -> bool
  ; ; (equal-elt? a) rend #t ssi «a» est égal à «elt»
  (define (equal-elt? a)
    (equal? a elt) )
  ; ; expression de (tous-egaux-a? L elt) :
  (verifient-tous? equal-elt? L) )

```

## Exercice II : typage d'une expression

Le but de cet exercice est de typer des expressions, complètement parenthésées et constantes. Dans un premier temps, nous considèrerons que les opérateurs sont des opérateurs binaires et que les expressions sont écrites en infixé. Plus précisément, le langage considéré comporte les constantes numériques (par exemple, 45, 67.8...), les constantes booléennes (`#t`, `#f`), les opérateurs arithmétiques binaires (`+`, `*`, `-`, `/`), les opérateurs logiques (`et`, `ou`) et les opérateurs de comparaison (`<`, `=`, `>`). On supposera que les comparaisons peuvent être effectuées sur des nombres ou des booléens (on prendra `#f` strictement inférieur à `#t`), mais pas entre un nombre et un booléen.

**Question 1** (4 points) :

Écrire une définition de la fonction `op-comparaison?` spécifiée par :

```
;; ; op-comparaison? : Symbole -> bool
;; ; (op-comparaison? s) rend #t ssi «s» est un opérateur de comparaison (<, =, >)
```

Par exemple :

```
(op-comparaison? '<) → #T
(op-comparaison? '+) → #F
(op-comparaison? 'a) → #F
```

```
(define (op-comparaison? s)
  (or (equal? s '<) (equal? s '=) (equal? s '>)))
```

Dans la suite, on pourra aussi faire appel aux fonctions `op-logique?` et `op-arithmétique?` de spécification :

```
;; ; op-logique? : Symbole -> bool
;; ; (op-logique? s) rend #t ssi «s» est un opérateur logique (et, ou)
```

```
;; ; op-arithmétique? : Symbole -> bool
;; ; (op-arithmétique? s) rend #t ssi «s» est un opérateur arithmétique (+, *, -, /)
```

On définit le type d'une expression de la façon suivante :

- une expression réduite à une constante numérique est de type *Numérique*,
- une expression réduite à une constante booléenne est de type *Booléen*,
- lorsque `op` est un opérateur arithmétique, une expression de la forme `(exp1 op exp2)` est
  - de type *Numérique* lorsque `exp1` et `exp2` sont de type *Numérique*,
  - mal typée sinon (autrement dit, lorsque `exp1` et `exp2` ne sont pas de type *Numérique* ou que l'une d'elles est mal typée),
- lorsque `op` est un opérateur logique, une expression de la forme `(exp1 op exp2)` est
  - de type *Booléen* lorsque `exp1` et `exp2` sont de type *Booléen*,
  - mal typée sinon (autrement dit, lorsque `exp1` et `exp2` ne sont pas de type *Booléen* ou que l'une d'elles est mal typée),
- lorsque `op` est un opérateur de comparaison, une expression de la forme `(exp1 op exp2)` est
  - de type *Booléen* lorsque `exp1` et `exp2` sont de même type,
  - mal typée sinon (autrement dit, lorsque `exp1` et `exp2` ne sont pas de même type ou que l'une d'elles est mal typée).

Voici des exemples d'expressions bien typées :

```
3 est de type Numérique
#f est de type Booléen
(3 + 5) est de type Numérique
(3 < (5 + 6)) est de type Numérique
(#f < #t) est de type Booléen
((3 < 5) et (3 < (5 + 6))) est de type Booléen
```

Voici des exemples d'expressions mal typées :


(3 + #f)  
 (#t < (5 + 6))  
 (3 et 5)  
 (5 + (3 + #f))

**Question 2** (4 points) :

Pour chacune des expressions précédentes, expliquer pourquoi elle est mal typée.

(3 + #f) : #f est de type *Booléen* alors que les opérandes de l'opérateur + doivent être de type *Numérique*.  
 (#t < (5 + 6)) : #t est de type *Booléen* et (5 + 6) est de type *Numérique* alors que les deux opérandes de l'opérateur > doivent être de même type  
 (3 et 5) : 3 et 5 sont de type *Numérique* alors que les opérandes de l'opérateur + doivent être de type *Booléen*.  
 (5 + (3 + #f)) : (3 + #f) n'est pas bien typé (cf. premier exemple).

Nous représentons ces expressions (bien formées mais bien ou mal typées) par des arbres binaires. Par exemple, l'expression ((3 \* (5 + 1)) + ((6 - 2) / 4)) est représentée par l'arbre



Le type `ArbreExprBinConst` est alors le type des arbres binaires qui représentent des expressions bien formées.

**Remarque :** dans toute la suite de l'exercice, on manipulera les arbres binaires **uniquement** à travers les fonctions de la barrière d'abstraction (cf. carte de référence). On pourra aussi faire appel aux fonctions `ab-feuille` et `ab-feuille?`, de spécification :

;; `ab-feuille` :  $\alpha \rightarrow \text{ArbreBinaire}[\alpha]$   
 ;; (`ab-feuille e`) rend la feuille d'étiquette «e»

;; `ab-feuille?` :  $\text{ArbreBinaire}[\alpha] \rightarrow \text{bool}$   
 ;; (`ab-feuille? B`) rend vrai ssi «B» est une feuille

**Question 3** (8 points) :

On considère le type `TypeExpression` égal à `{Numerique, Booleen}`, la valeur `Numerique` représentant le type des expressions numériques et la valeur `Booleen` représentant le type des expressions booléennes.

Écrire une définition de la fonction `abe-type` spécifiée par :

```
;; ; abe-type : ArbreExprBinConst -> TypeExpression
;; ; ERREUR lorsque l'expression représentée par «expr» n'est pas bien typée
;; ; (abe-type expr) rend le type de l'expression représentée par «expr»
```

```
(define (abe-type expr)
  (if (ab-feuille? expr)
      (cond ((number? (ab-etiquette expr)) 'Numerique)
            ((boolean? (ab-etiquette expr)) 'Booleen)
            (else
             (erreur 'abe-type
                    "les feuilles doivent être des nombres ou des booléens")))
      (let ((op (ab-etiquette expr))
            (t-gauche (abe-type (ab-gauche expr)))
            (t-droit (abe-type (ab-droit expr))))
        (cond ((op-comparaison? op)
               (if (equal? t-gauche t-droit)
                   'Booleen
                   (erreur 'abe-type "opérateur de comparaison" ) )
               ((op-logique? op)
                (if (and (equal? t-gauche 'Booleen) (equal? t-droit 'Booleen))
                    'Booleen
                    (erreur 'abe-type "opérateur logique" ) )
                ((op-arithmetique? op)
                 (if (and (equal? t-gauche 'Numerique) (equal? t-droit 'Numerique))
                     'Numerique
                     (erreur 'abe-type "opérateur arithmétique" ) )
                 (else
                  (erreur 'abe-type "opérateur inconnu" ) ) ) ) ) ) ) )
```

Nous considérons maintenant que les opérateurs arithmétiques et logiques peuvent avoir un nombre quelconque d'arguments et nous représentons les expressions par des arbres généralisés. Le type des arbres représentant de telles expressions est nommé `ArbreExprGenConst`.

Le but de la suite de l'exercice est de définir la fonction `age-type` spécifiée par :

```
;; ; age-type : ArbreExprGenConst -> TypeExpression
;; ; ERREUR lorsque l'expression représentée par «expr» n'est pas bien typée
;; ; (age-type expr) rend le type de l'expression représentée par «expr»
```

**Question 4** (4 points) :

En utilisant la fonction `age-type`, écrire une définition de la fonction `tous-type?` spécifiée par :

```
;; ; tous-type? : LISTE[ArbreExprGenConst] * TypeExpression -> bool
;; ; ERREUR lorsque l'expression représentée par un des éléments de «LA» est mal typée.
;; ; (tous-type? LA t) rend #t ssi tous les éléments de la liste d'arbres
;; ; d'expression «LA» sont de type «t»
```

```
(define (tous-type? LA t)
  (if (pair? LA)
      (and (equal? (age-type (car LA)) t)
            (tous-type? (cdr LA) t) )
      #t ) )
```

Une autre définition qui utilise l'itérateur `verifient-tous?` du premier exercice :

```
(define (tous-type1? LA t)
  ;; ; type-t? : ArbreExprGenConst -> bool
  ;; ; (type-t? expr) rend #t ssi «expr» est de type «t»
  (define (type-t? expr)
    (equal? (age-type expr) t))

  ;; ; expression de (tous-type1? LA t):
  (verifient-tous? type-t? LA))
```

**Question 5** (8 points) :

Écrire une définition de la fonction `age-type`. Indication : on peut utiliser la fonction `tous-type?`.

*;; ; age-type : ArbreExprGenConst -> TypeExpression*

*;; ; ERREUR lorsque l'expression représentée par «expr» n'est pas bien typée*

*;; ; (age-type expr) rend le type de l'expression représentée par «expr»*

```
(define (age-type expr)
  (if (ag-feuille? expr)
      (cond ((number? (ag-etiquette expr)) 'Numerique)
            ((boolean? (ag-etiquette expr)) 'Booleen)
            (else
             (erreur 'age-type
                     "les feuilles doivent être des nombres ou des booléens")))
      (let ((op (ag-etiquette expr))
            (args (ag-foret expr)))
        (cond ((op-comparaison? op)
               (if (equal? (age-type (car args)) (age-type (cadr args)))
                   'Booleen
                   (erreur 'age-type "opérateur de comparaison") ) )
              ((op-logique? op)
               (if (tous-type? args 'Booleen)
                   'Booleen
                   (erreur 'age-type "opérateur logique") ) )
              ((op-arithmetique? op)
               (if (tous-type? args 'Numerique)
                   'Numerique
                   (erreur 'age-type "opérateur arithmétique") ) )
              (else
               (erreur 'age-type "opérateur inconnu")) ) ) ) ) )
```